

... SPECTRAL PLOT**PURPOSE**

Generates one of the following types of spectral plots:

1. spectral plot (= autospectral plot)
2. cross-spectral plot
3. co-spectral plot
4. quadrature spectral plot
5. coherency spectral plot
6. amplitude spectral plot
7. phase spectral plot
8. gain spectral plot
9. argand spectral plot

DESCRIPTION

A spectral plot is a graphical data analysis technique for examining frequency-domain models for a single equi-spaced time series or for two equi-spaced time series. It is used to assess autocorrelation and cyclic structure. The spectral power function is a smoothed Fourier transform of the autocovariance function. An equi-spaced time series is one in which the distance between adjacent points is constant. The spectral estimate (also called power) at a given frequency for a single discrete time series x_t is:

$$\bar{C}_{xx}(f) = 2\Delta \left(c_{xx}(0) + 2 \sum_{k=1}^{L-1} c_{xx}(k)w(k) \cos(2\pi fk\Delta) \right) \quad (\text{EQ 2-20})$$

In the above equation, c_{xx} is the autocovariance function evaluated at the lag k , Δ is the distance between adjacent values (DATAPLOT always uses a value of 1), L defines how many lags of the autocovariance function to use, and f is the frequency. The $w(k)$ is the window function (which specifies the type of smoothing). The window function, the number of lags to use in the autocovariance function, and the formulas for the other variations of the spectral plot are discussed further in NOTE sections below. The autospectral plot consists of:

Vertical axis = power (= variable contribution);

Horizontal axis = frequency (cycles per observation).

The frequency is measured in cycles per unit time where unit time is defined to be the distance between adjacent points. A frequency of 0 corresponds to an infinite cycle while a frequency of 0.5 corresponds to a cycle of 2 data points. Equi-spaced time series are inherently limited to detecting frequencies between 0 and 0.5.

From a data analysis point of view, the type of structure in the autocorrelation plot indicates the location of peaks in the spectral plot. The peaks in the spectral plot also indicate the dominant frequency for underlying cyclic models. Once the dominant peaks have been identified, the next step is typically to use other time series analysis techniques (such as the complex demodulation phase plots) to determine if this frequency is constant over the entire domain of the data, or to carry out a nonlinear fit with an underlying cyclic model (see the documentation for the COMPLEX DEMODULATION PLOT command).

SYNTAX 1

SPECTRAL PLOT <y1> <SUBSET/EXCEPT/FOR qualification>

where <y1> is a response variable;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for a single time series.

SYNTAX 2

<keyword> PLOT <y1> <y2> <SUBSET/EXCEPT/FOR qualification>

where <keyword> is one of the following:

CROSS-SPECTRAL
 COSPECTRAL
 QUADRATURE SPECTRAL
 COHERENCY SPECTRAL
 AMPLITUDE SPECTRAL
 PHASE PSECTRAL

GAIN SPECTRAL,
ARGAND SPECTRAL;

<y1> is the first response variable;

<y2> is the second response variable;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for two time series.

EXAMPLES

SPECTRAL PLOT Y

SPECTRAL PLOT Y2 SUBSET Y2 > 2

CROSS-SPECTRAL PLOT Y1 Y2

CO-SPECTRAL PLOT Y1 Y2

QUADRATURE SPECTRAL PLOT Y1 Y2

COHERENCY SPECTRAL PLOT Y1 Y2

AMPLITUDE SPECTRAL PLOT Y1 Y2

PHASE SPECTRAL PLOT Y1 Y2

ARGAND SPECTRAL PLOT Y1 Y2

NOTE 1

Missing values are not allowed. It is also common to remove trends by differencing ($x_t - x_{t-1}$) or to apply some other type of filter before generating the spectral plot.

NOTE 2

Different time series texts present the equations for the various spectral plots in slightly different forms. However, these forms should be mathematically equivalent. DATAPLOT uses the definitions from the Jenkins and Watts book (see REFERENCE below). The Bloomfield text (see REFERENCE below) uses slightly different, although mathematically equivalent, equations.

NOTE 3

The spectral plot is essentially a “smoothed” periodogram where the smoothing is done in the frequency domain. The periodogram is the Fourier transform of the autocovariance function while the autospectral plot is the smoothed Fourier transform of the autocovariance function. The periodogram can be normalized to generate a spectral density by using the autocorrelation function rather than the autocovariance function in the computational formulas (see the documentation for the PERIODOGRAM command). In a similar manner, the various spectral plots can be normalized by using the autocorrelation function rather than the autocovariance function. However, DATAPLOT does not currently provide a mechanism for specifying that the normalized estimates be used.

NOTE 4

The smoothing is determined by the choice of the window function. The 4 most common choices are the rectangular window, the Bartlett window, the Tukey window, and the Parzen window. These are all described in table 6.5 of the Jenkins and Watts text (see REFERENCE below). DATAPLOT uses the Tukey window. At this time, it does not allow the specification of an alternate window type. The formula for the Tukey window is:

$$w(u) = \frac{1}{2} \left(1 + \cos \left(\frac{\pi u}{M} \right) \right) \quad (|u| \leq M) \quad (\text{EQ 2-21})$$

where M is the number of lags to compute the autocorrelation for and w(u) is zero if u is outside the given region.

NOTE 5

The number of lags is determined automatically. By default, the number of lags is (N is the number of points in the time series):

(N/4) - 1	N > 32
(N/2) - 1	16 < N <= 32
N - 1	N <= 16

If you want to over-ride the default value, enter one of the following commands:

```
LET LAGS = <value>
LET LAG = <value>
LET NUMLAG = <value>
```

NOTE 6

The formula for the spectral plot above was for the autospectral plot. This section gives the formula for the other spectral plots.

The sample autocovariance functions are computed with the standard formulas (call these c_{11} and c_{22} for the first and second series respectively). The autospectral estimates are computed for the first and second series using the formula given above (call these \bar{C}_{11} and \bar{C}_{22} respectively).

In the formulas below, L is the number of lags of the autocovariance function and F is the maximum frequency (this will always be 0.5 since DATAPLOT assumes the distance between 2 points is one unit of time). The $w(f)$ is the Tukey window discussed above.

The next step is to calculate the cross covariance functions:

$$c_{12}(k) = \frac{1}{N} \sum_{t=1}^{N-k} (x_{1t} - \bar{x}_1)(x_{2t+k} - \bar{x}_2) \quad 0 \leq k \leq L-1 \quad (\text{EQ 2-22})$$

$$c_{12}(-k) = \frac{1}{N} \sum_{t=1}^{N-k} (x_{1t+k} - \bar{x}_1)(x_{2t} - \bar{x}_2) \quad 0 \leq k \leq L-1 \quad (\text{EQ 2-23})$$

From these, the even and odd cross covariance functions are calculated:

$$l_{12}(k) = \frac{1}{2}(c_{12}(k) + c_{12}(-k)) \quad 0 \leq k \leq L-1 \quad (\text{EQ 2-24})$$

$$q_{12}(k) = \frac{1}{2}(c_{12}(k) - c_{12}(-k)) \quad 0 \leq k \leq L-1 \quad (\text{EQ 2-25})$$

The co-spectral estimate is:

$$\bar{L}_{12}(f) = 2 \left(l_{12}(0) + 2 \sum_{k=1}^{L-1} l_{12}(k) w(k) \cos\left(\frac{\pi f k}{F}\right) \right) \quad 0 \leq f \leq F \quad (\text{EQ 2-26})$$

The quadrature spectral estimate is (it is always zero for frequencies 0 and F):

$$\bar{Q}_{12}(f) = 4 \sum_{k=1}^{L-1} q_{12}(k) w(k) \sin\left(\frac{\pi f k}{F}\right) \quad 0 \leq f \leq F \quad (\text{EQ 2-27})$$

The cross spectrum is a plot of the co-spectral plot and the quadrature spectrum plot on the same graph.

The cross amplitude spectral estimate is:

$$\bar{A}_{12}(f) = \sqrt{\bar{L}_{12}^2 + \bar{Q}_{12}^2} \quad 0 \leq f \leq F \quad (\text{EQ 2-28})$$

The phase spectral estimate is:

$$\bar{F}_{12}(f) = \arctan\left(\frac{\bar{Q}_{12}}{\bar{L}_{12}}\right) \quad 0 \leq f \leq F \quad (\text{EQ 2-29})$$

The coherency spectral estimate is:

$$\bar{K}_{12}(f) = \frac{\bar{A}_{12}}{\bar{C}_{11}(f)\bar{C}_{22}(f)} \quad 0 \leq f \leq F \quad (\text{EQ 2-30})$$

The gain spectral estimate is (this is the cross amplitude divided by the autospectrum of the first series):

$$\bar{G}_{12}(f) = \frac{\sqrt{\bar{L}_{12}^2 + \bar{Q}_{12}^2}}{\bar{C}_{11}(f)} \quad 0 \leq f \leq F \quad (\text{EQ 2-31})$$

The argand spectral plot consist of:

Vertical axis = the co-spectral estimate divided by the autospectral estimate of the first series;

Horizontal axis = the quadrature spectral estimate divided by the autospectral estimate of the second series.

NOTE 7

The frequency is computed at a discrete number of points. DATAPLOT computes it at $(N/2)$ equally spaced points where N is the number of points in the time series. However, it uses 120 frequencies as a lower bound and 1000 frequencies as an upper bound.

NOTE 8

DATAPLOT uses the algorithms on pages 310-312 and 418-420 of the Jenkins and Watts book (see REFERENCE). Although spectral estimates can be computed in terms of the fast Fourier transform, DATAPLOT does not use that method.

NOTE 9

Spectral plots are often drawn with log scales to provide better resolution (enter YLOG ON). Some analysts prefer to draw the plot as a solid connected lines while others prefer to draw it with spikes. Either way is straightforward to generate with the proper settings for the LINES and SPIKES commands.

DEFAULT

None

SYNONYMS

SPECTRUM is a synonym for SPECTRAL.

The word PLOT is optional in the various SPECTRAL PLOT commands (e.g., SPECTRUM Y, CROSS-SPECTRUM Y1 Y2).

AUTO SPECTRAL PLOT is a synonym for SPECTRAL PLOT.

RELATED COMMANDS

PERIODOGRAM	=	Generates a periodogram plot.
CORRELATION PLOT	=	Generates a correlation plot.
COMPLEX DEMOD PLOT	=	Generates a complex demodulation plot.
LAG PLOT	=	Generates a lag plot.
PLOT	=	Generates a data or function plot.
4-PLOT	=	Generates 4-plot univariate analysis.
CHARACTERS	=	Sets the types for plot characters.
LINES	=	Sets the types for plot lines.
LOG	=	Sets the log switches on/off.
SPIKES	=	Sets the on/off switches for plot spikes.
SUMMARY	=	Generates a table of summary statistics.
LET	=	Generates sin/cos transformations (plus much more).
FIT	=	Carries out a least squares fit.

REFERENCE

“Spectral Analysis and Its Applications,” Jenkins and Watts, Holden-Day, 1968 (chapters 6-10).

“Spectral Analysis of Economic Time Series,” Granger and Hatanaka, Princeton University Press, 1964 (pp. 77-79).

“Fourier Analysis of Time Series,” Peter Bloomfield, John Wiley and Sons, 1976.

APPLICATIONS

Frequency Time Series Analysis


IMPLEMENTATION DATE

Pre-1987

PROGRAM 1

```

. THIS SAMPLE PROGRAM READS THE FILE LEW.DAT IN THE DATAPLOT
. REFERENCE DIRECTORY. THESE DATA ARE BEAM DEFLECTION DATA.
.
SKIP 25
READ LEW.DAT DEFLECT
.
TITLE AUTOMATIC
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
LET A = MEAN DEFLECT
LINE BLANK; SPIKE ON; SPIKE BASE A
PLOT DEFLECT
.
SPIKE BASE 0
LINE BLANK SOLID DOT DOT
YILABEL AUTOCORRELATION
XILABEL LAG
AUTOCORRELATION PLOT DEFLECT
.
YILABEL POWER
XILABEL FREQUENCY
SPECTRAL PLOT DEFLECT
.
YLOG ON
LINE SOLID
SPIKE OFF
SPECTRAL PLOT DEFLECT
.
END OF MULTIPLY
    
```



PROGRAM 2

```
. READ THE FILE DUTTON.DAT IN THE DATAPLOT REFERENCE DIRECTORY.
SKIP 25
READ DUTTON.DAT X Y1 Y2 Y3
.
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE AUTOMATIC; LINES SOLID DASH
PLOT Y1 Y3 VS X
Y1LABEL POWER; X1LABEL FREQUENCY
SPECTRAL PLOT Y1
SPECTRAL PLOT Y3
LINE BLANK SOLID DOT DOT DOT DOT; SPIKE ON; SPIKE BASE 0
Y1LABEL CORRELATION; X1LABEL LAG
CROSS CORRELATION PLOT Y1 Y3
. PAGE 2
MULTIPLY 4 2
SPIKE OFF; LINES SOLID DASH
YMINIMUM 0; YTIC OFFSET 1 0; Y1LABEL POWER; X1LABEL FREQUENCY
CROSS-SPECTRAL PLOT Y1 Y3
COSPECTRAL PLOT Y1 Y3
AMPLITUDE SPECTRAL PLOT Y1 Y3
YMINIMUM; YTIC OFFSET 0 0
QUADRATURE SPECTRAL PLOT Y1 Y3
COHERENCY SPECTRAL PLOT Y1 Y3
PHASE SPECTRAL PLOT Y1 Y3
GAIN SPECTRAL PLOT Y1 Y3
Y1LABEL CO-SPECTRUM/SPECTRUM 1
X1LABEL QUADRATURE SPECTRUM/SPECTRUM 2
ARGAND SPECTRAL PLOT Y1 Y3
END OF MULTIPLY
```

